

Solutions to Selected Exercises

To derive the maximum benefit from these exercises, make an honest effort to create the Maple code before consulting the solution offered here.

Part I. The Maple Worksheet

Section 1. Execution Groups: Input/Output

2.* Simplify the square root of 19,220 by entering and executing

```
sqrt(19220);
```

Then obtain the complete integer factorization of 19220 by entering and executing

```
ifactor(19220);
```

Read **ifactor** as "integer factors of".

```
> sqrt(19220);
```

$$62\sqrt{5}$$

```
> ifactor(19220);
```

$$(2)^2 (5) (31)^2$$

3.* Obtain an integral of the expression

$$e^{(3x)} \cos(2x) \sin(4x).$$

Enter it into the **int** procedure as follows

```
int( exp(3*x)*cos(2*x)*sin(4*x), x);
```

```
> int( exp(3*x)*cos(2*x)*sin(4*x), x);
```

$$-\frac{1}{15} e^{(3x)} \cos(6x) + \frac{1}{30} e^{(3x)} \sin(6x) + \frac{3}{13} e^{(3x)} \cos(2x) - \frac{9}{26} e^{(3x)} \sin(2x) + \frac{2}{13} e^{(3x)} (3 \sin(2x) - 2 \cos(2x))$$

Section 2. Entering blocks of text

18.* Save your worksheet (**Command-S**) then print it.

See the pdf file entitled My_2nd_wrksht.pdf.

Part II. Calculations and Calculus with Maple

Section 1. Getting Started: Maple as a Calculator

6. Use the sequence operator, \$, to make the following sequences.

a. 2, 4, 6, 8, 10, 12

b. 20, 40, 60, ..., 260 Hint. $20 + 20*k$ \$ $k=$

c. The 30 prime numbers starting with 11 and ending with 139. Hint. Use **ithprime**

7.* Add the numbers in the sequence in 6 c.. Hint. If you just did 6 c, then execute **add(k,k=%)**.

```
> ithprime(k) $ k=5..33;
11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67, 71, 73, 79, 83, 89, 97, 101, 103, 107, 109, 127, 131, 137
```

```
> add(k,k=%);
```

1971

10. Use seq to obtain the sequence of the cubes of the numbers in the list [2, 5, 6, 9, 12, 44]. Add the numbers in the sequence of cubes and then obtain the prime factorization of the sum.

What do you notice about the prime factorization? (Hint. The integer 87990 is called "square free".)

11.* Obtain the prime factorization of the product of the integers in the sequence of cubes described in Exercise 10.

```
> seq( k^3, k=[2,5,6,9,12,44]);
8, 125, 216, 729, 1728, 85184
```

```
> mul(k,k=%);
ifactor(%);
```

23178378313728000

$(2)^{18} (3)^{12} (5)^3 (11)^3$

12. Use the factor procedure to factor the following polynomial expressions.

a. $x^3 - x^2 + x - 1$

b. $x^7 - x^6 + x^5 - \dots - 1$ Hint. Enter this as **-add((-x)^k,k=0..7)** then execute **factor(%)**.

15.* Use solve to obtain the zeros of the polynomials in Exercise 12.

```
> x^3 - x^2 + x - 1;
solve(%);
```

$$x^3 - x^2 + x - 1$$

1, I, -I

```
> -add((-x)^k,k=0..7);
solve(%);
```

$$-1 + x - x^2 + x^3 - x^4 + x^5 - x^6 + x^7$$

$1, I, -I, \frac{1}{2}\sqrt{2} + \frac{1}{2}I\sqrt{2}, -\frac{1}{2}\sqrt{2} + \frac{1}{2}I\sqrt{2}, -\frac{1}{2}\sqrt{2} - \frac{1}{2}I\sqrt{2}, \frac{1}{2}\sqrt{2} - \frac{1}{2}I\sqrt{2}$

16.* Add the zeros of the polynomial in 12 a and the zeros of the polynomial in 12 b.

```
> solve(x^3 - x^2 + x - 1);
add(k,k=%);
```

1, I, -I

1

> solve(-add((-x)^k,k=0..7));
add(k,k=%);

$1, I, -I, \frac{1}{2}\sqrt{2} + \frac{1}{2}I\sqrt{2}, -\frac{1}{2}\sqrt{2} + \frac{1}{2}I\sqrt{2}, -\frac{1}{2}\sqrt{2} - \frac{1}{2}I\sqrt{2}, \frac{1}{2}\sqrt{2} - \frac{1}{2}I\sqrt{2}$

1

Section 2. Symbolics: Equations and Assignments

6.* Plot the expression

$$x^3 - x^2 + 0.05x + \cos(x) - 0.7$$


with the entry

`y := x^3 - x^2 + 0.05*x + cos(x) - 0.7; plot(y, x=-2..2, -2..2);`

Enter `fsolve(y=0,x)` to see which zero fsolve finds.

Using the graph as a guide, obtain an approximation to the largest positive zero using `fsolve(y=0,{x},a..b)`.

> `y := x^3 - x^2 + 0.05*x + cos(x) - 0.7; plot(y, x=-2..2, -2..2);`
 $y := x^3 - x^2 + 0.05x + \cos(x) - 0.7$



> `fsolve(y=0,x);`

0.6180709010

> `fsolve(y=0,x,1..2);`

1.189778784

9. Graph the function

$$y = \cos(x^2) - \sin(2x)$$

over the interval from $x = 0$ to $x = 2$. Define yp as its derivative and find the zero of yp near $x = 1.5$. Name it $xmin$. Then find the minimum value of y over this interval using `eval(y,x=xmin)`. Name it $ymin$.

Use the following entry to plot the graph and the low point.

```
plot( [y, [ [xmin,ymin] ] ], x=0..2, style=[line,point]);
```


10.* Find the area of the region between the graph in Exercise 9 and the x-axis.

Hint: Numerically integrate the absolute value of y from x = 0 to x = 2 via the entry

```
evalf( Int( abs(y), x=0..2) );
```

```
> y := cos(x^2) - sin(2*x);  
plot(y, x=0..2);
```

$y := \cos(x^2) - \sin(2x)$



```
> evalf( Int( abs(y), x=0..2) );
```

0.9147700913

Section 3. Functions as Transformations

5.* An animation.

The tangent line plot in Exercise 3 can easily be animated as follows.

First define the function $T(a,x)$ whose value at (a,x) is the formula for the tangent line to the graph of f at $(a,f(a))$.

```
T := (a,x) -> f(a) + g(a)*(x - a);
```

Then use

```
plots[animate](plot,[ [f(x), [t,T(a,t),t=a-0.5..a+0.5] ],x=-1..1,-1..2,color=[red,blue],a = -1..1);
```


Once the plot appears, click on it with the mouse and the context bar becomes a row of video controls; enjoy.

Read the Help page for plots[animate].

```
> f := x -> cos(x) - x*tan(x): g := D(f):  
T := (a,x) -> f(a) + g(a)*(x - a);
```

$T := (a, x) \rightarrow f(a) + g(a)(x - a)$

```
> plots[animate](plot,[ [f(x), [t,T(a,t),t=a-0.5..a+0.5] ], x=-1..1,  
-1..2, color=[red,blue] ], a = -1..1);
```



8.* Using unapply and Matrix.

Convert the second solution in Exercise 6 into a function f using

```
f := unapply(rhs(soln),x);
```

Evaluate the solution at x = 0, 0.2, 0.4, ... , 1.0 to 4 digits as follows

```
evalf[4]( f(0.2*k) $ k=0..5 );
```

Make a table of function values with the following Matrix entry

```
Matrix( [ [x, 0.2*k $ k=0..5] , [ 'f(x)', % ] ] );
```

```
> restart: deqn := diff(y(x),x) + x*y(x) = x:
soln := dsolve( {deqn,y(0)=0} ):
f := unapply(rhs(soln),x);
```

$$f := x \rightarrow 1 - e^{\left(-\frac{1}{2}x^2\right)}$$

```
> evalf[4](f(0.2*k) $ k=0..5);
0., 0.0198, 0.0769, 0.1647, 0.2739, 0.3935
```

```
> Matrix( [ [x, 0.2*k $ k=0..5] , [ 'f(x)', % ] ] );
```

| | | | | | | |
|------|----|--------|--------|--------|--------|--------|
| x | 0. | 0.2 | 0.4 | 0.6 | 0.8 | 1.0 |
| f(x) | 0. | 0.0198 | 0.0769 | 0.1647 | 0.2739 | 0.3935 |

Part III. First Order Ordinary Differential Equations

Section 1. Entering, Solving, Plotting

7.* Enter the differential equation $y' + y = \cos(t)$ with the name DE. Obtain the solution satisfying the generic initial condition $y(t_0) = y_0$. Name it soln.

```
> DE := diff(y(t),t) + y(t) = cos(t);
soln := dsolve( {DE, y(t0)=y0} );
```

$$DE := \left(\frac{d}{dt} y(t) \right) + y(t) = \cos(t)$$

$$\text{soln} := y(t) = \frac{1}{2} \cos(t) + \frac{1}{2} \sin(t) - \frac{e^{(-t)} \left(-y_0 + \frac{1}{2} \cos(t_0) + \frac{1}{2} \sin(t_0) \right)}{e^{(-t_0)}}$$

a. Use the entry `eval(soln, t=t0)` to verify that soln satisfies the initial condition.

> `eval(soln, t=t0);`

$$y(t_0) = y_0$$


b. Use `unapply` to make `rhs(soln)` into a function of `t`, `t0`, and `y0` named `phi` (see page 35 in Part III, Section 1 of the manual).

> `phi := unapply(rhs(soln), t, t0, y0);`

$$\phi := (t, t_0, y_0) \rightarrow \frac{1}{2} \cos(t) + \frac{1}{2} \sin(t) - \frac{e^{(-t)} \left(-y_0 + \frac{1}{2} \cos(t_0) + \frac{1}{2} \sin(t_0) \right)}{e^{(-t_0)}}$$


c. Use `phi` to plot some solutions starting at points evenly spaced on the y axis.

> `plot([[t, phi(t, 0, y0), t=0..10] $ y0=-2..2]);`




d. Use `phi` to plot some solutions starting at points evenly spaced on the t axis.

> `plot([[t, phi(t, t0, 0), t=t0..10] $ t0=-2..2]);`




e. Use `phi` to plot the solutions starting at points evenly spaced around the unit circle in the style of the two plots on page 35 of the manual. That is, one picture runs time forward, another runs time backward.

```
> Points := [cos(k*Pi/6),sin(k*Pi/6)] $ k=0..11:
plot( [ [t,phi(t,cos(k*Pi/6),sin(k*Pi/6)),t=cos(k*Pi/6)..2]
$ k=0..11, [ Points ] ], style=[line$12,point],
color=[red$12,black], view=[-2..2,-2..2] );
```



```
> plot( [ [t,phi(t,cos(k*Pi/6),sin(k*Pi/6)),t=-2..cos(k*Pi/6)]
$ k=0..11, [ Points ] ], style=[line$12,point],
color=[red$12,black], view=[-2..2,-2..2] );
```



Section 2. Working with Solutions: Modeling

3.* The following problem is adapted from Ledder, Chapter 1, Section 1.

Suppose you borrow \$12,000 to buy a car. The loan is to be paid in 60 equal monthly installments at an interest rate of 5% per year.

a. Assume the payments are actually made continuously at whatever rate is needed to pay off the loan in 60 months. Determine the continuous rate per month that would be required.

```
> DE := diff(P(t),t) = 0.05*P(t) - Q;
soln := dsolve( {DE,P(0)=12000} );
```

$$DE := \frac{d}{dt} P(t) = 0.05 P(t) - Q$$

$$soln := P(t) = 20 Q + e^{\left(\frac{1}{20} t\right)} (12000 - 20 Q)$$

```
> eqn := subs(P(t)=0,t=5,soln);
Q := solve(eqn,Q);
```

$$eqn := 0 = 20 Q + e^{\left(\frac{1}{4}\right)} (12000 - 20 Q)$$

$$Q := \frac{600 e^{\left(\frac{1}{4}\right)}}{-1 + e^{\left(\frac{1}{4}\right)}}$$

```
> Monthly_Rate = evalf(Q/12)*dollars;
Monthly_Rate = 226.0405830 dollars
```

b. Compare the answer to part a to the answer if 60 equal monthly payments are made at a constant annual interest rate of 5% applied to the outstanding balance. In other words, the first payment, due one month after the loan is made, would be

$$12000 \left(e^{\left(\frac{0.05}{12}\right)} - 1 \right) + P_1$$

where P[1] is the amount that is put towards reducing the principal (the \$12000) in the first month. The second payment is

$$(12000 - P_1) \left(e^{\left(\frac{0.05}{12}\right)} - 1 \right) + P_2$$

where P[2] is the amount put towards reducing the principal in the second month. The nth payment is

$$\left(12000 - \left(\sum_{k=1}^{n-1} P_k \right) \right) \left(e^{\left(\frac{0.05}{12}\right)} - 1 \right) + P_n .$$

Hint for b. The second payment is supposed to equal to the first. Therefore, the following equation must be satisfied.

```
> eqn := 12000*(exp(0.05/12)-1)+P[1] =
(12000 - P[1])* (exp(0.05/12)-1) + P[2];
eqn := 50.10431 + P1 = 50.10430800 - 0.004175359 P1 + P2
```

This determines P[2] in terms of P[1].

```
> P[2] := solve(eqn,P[2]);
P2 := 0.000002000000000 + 1.004175359 P1
```

Create a for..do loop that calculates P[3], P[4], ... , P[60] in terms of P[1], then find P[1] using the fact

that

$$\sum_{k=1}^{60} P_k = 12000$$

```
> for n from 3 to 60
  do
    eqn := (12000-add(P[k],k=1..n-2))*(exp(0.05/12) - 1) + P[n-1] =
           (12000-add(P[k],k=1..n-1))*(exp(0.05/12) - 1) + P[n]:
    P[n] := solve(eqn,P[n]):
  end do:
unassign('n');

> add(P[k],k=1..60) = 12000:
P[1] := solve(%,P[1]);
```

$$P_1 := 176.4078443$$

```
> Check;
for n from 1 to 60 by 19
  do
    payment,n>equals,(12000-add(P[k],k=1..n-1))*(exp(0.05/12) - 1) +
    P[n];
  end do;
unassign('n');
```

Check

$$\text{payment, 1, equals, 226.5121523}$$

$$\text{payment, 20, equals, 226.5121541}$$

$$\text{payment, 39, equals, 226.5121538}$$

$$\text{payment, 58, equals, 226.5121537}$$

P[1] is the first payment towards the principal. The 60th payment is almost all towards the principal.


```
> Last_Principal_payment=P[60];
Last_Principal_payment = 225.5703165
```

Section 3. Slope Fields: DEplot

2.* Use DEplot to make a nice-looking slope field for the autonomous equation $y' = \sin(y)$. (Use the window $t=-6..6, y=-6..6$.)


```
> with(DEtools):
> DE := diff(y(t),t) = sin(y(t));
DEplot( DE, y(t), t=-6..6, y=-6..6, arrows=line, dirgrid=[17,17] );
```

$$DE := \frac{d}{dt} y(t) = \sin(y(t))$$



a. Put some solution curves into the plot by adding `{ [y(0)=k] $ k=-6..6 }, linecolor=blue`. Comment on the relationship between one curve and the next.

```
> DEplot( DE, y(t), t=-6..6, y=-6..6, arrows=line, dirgrid=[17,17],
          { [y(0)=k] $ k=-6..6 }, linecolor=blue);
```



Each curve is a horizontal translate of the other.

b. Obtain the general solution using `dsolve`.

```
> dsolve( DE );
```

$$y(t) = \arctan \left(\frac{2 e^t _CI}{1 + e^{(2t)} _CI^2}, \frac{-e^{(2t)} _CI^2 + 1}{1 + e^{(2t)} _CI^2} \right)$$

c. Use `dsolve` to obtain the solution satisfying the initial condition $y(0) = 1$. What is the value of this solution when $t = 1$? Get the exact value and an approximation. If the solution is called "soln" use

```
eval(soln,t=1); evalf(%);
```

```
> soln := dsolve( {DE,y(0)=1} );
```

$$\text{soln} := y(t) = \arctan \left(\frac{2 e^t \sin(1) (-1 + \cos(1))}{-\sin(1)^2 - 2 (e^t)^2 + 2 (e^t)^2 \cos(1) + (e^t)^2 \sin(1)^2} \right),$$

$$- \frac{2 (e^t)^2 \cos(1) + \sin(1)^2 + (e^t)^2 \sin(1)^2 - 2 (e^t)^2}{-\sin(1)^2 - 2 (e^t)^2 + 2 (e^t)^2 \cos(1) + (e^t)^2 \sin(1)^2}$$

> `eval(soln,t=1); evalf(%);`

$$y(1) = -\arctan \left(\frac{2 e \sin(1) (-1 + \cos(1))}{2 (e)^2 \cos(1) + \sin(1)^2 + (e)^2 \sin(1)^2 - 2 (e)^2} \right) + \pi$$

$$y(1) = 1.956294970$$

d. Comment on the long-term behavior of solutions to this differential equation.

The slope field suggests that as t approaches infinity, all solutions have a horizontal asymptote. That is, the solutions have a finite limit (depending on the initial value) as t approaches infinity. The arctan function in the solution formula confirms this observation. One can also appeal to the fact that the values $y = k\pi$ are stationary points on the phase line. All solutions are either attracted to $k\pi$ or repelled from $k\pi$ for some k .

Section 4. Approximate Solutions

7. Modify the ModEuler procedure to make a procedure that implements the classical Runge-Kutta algorithm (see Ledger, Section 2.6). Test in by applying it to the problems described in Exercises 2 - 6. (Copy and Paste).

```
> RK := proc(f,t0,y0,h,N)
  local n, k, T, Y;
  T[0] := t0;
  Y[0] := y0;
  for n from 0 to N-1
    do
      T[n+1] := T[n] + h;
      k[1] := f(T[n],Y[n]);
      k[2] := f(T[n] + 0.5*h,Y[n] + 0.5*k[1]*h);
      k[3] := f(T[n] + 0.5*h,Y[n] + 0.5*k[2]*h);
      k[4] := f(T[n+1],Y[n] + k[3]*h);
      Y[n+1] := Y[n] + 1/6*(k[1]+2*k[2]+2*k[3]+k[4])*h;
    end do;
  [ [T[m],Y[m]] $ m=0..N ]
end proc;
```

```
RK := proc(f, t0, y0, h, N)
local n, k, T, Y;
  T[0] := t0;
  Y[0] := y0;
  for n from 0 to N - 1 do T[n + 1] := T[n] + h;
    k[1] := f(T[n], Y[n]);
```

```

k[2] := f(T[n] + 0.5*h, Y[n] + 0.5*k[1]*h);
k[3] := f(T[n] + 0.5*h, Y[n] + 0.5*k[2]*h);
k[4] := f(T[n + 1], Y[n] + k[3]*h);
Y[n + 1] := Y[n] + 1/6*(k[1] + 2*k[2] + 2*k[3] + k[4])*h;
end do;
['$'([T[m], Y[m]], m = 0 .. N)];
end proc;

```

Test on Exercise 5.

```

> DE := diff(y(t),t) = cos(t)*y(t);
f := (t,y) -> cos(t)*y;

```

$$DE := \frac{d}{dt} y(t) = \cos(t) y(t)$$

$$f := (t, y) \rightarrow \cos(t) y$$

```

> soln := dsolve( {DE, y(0)=1} );
g := unapply(rhs(soln),t);

```


$$soln := y(t) = e^{\sin(t)}$$

$$g := \exp @ t \rightarrow \sin(t)$$

```

> L1 := RK(f,0,1,0.5,20);
plot( [L1,g(t),L1], t=0..10, y=0..3, style=[line$2,point],
color=[red,blue,black]);


```



```

> L2 := RK(f,0,1,0.25,40);
plot( [L2,g(t),L2], t=0..10, y=0..3, style=[line$2,point],
color=[red,blue,black]);

```



```
> Matrix([ [t,0.5*k$ k=0..5],
            ['h=0.5',evalf[6]('L1[k,2]'$k=1..6)],
            ['h=0.25',evalf[6]('L2[2*k+1,2]'$k=0..5)],
            ['g(t)',evalf[6](g(0.5*k)$k=0..5)] ] );
```

| | | | | | | |
|------------|----|---------|---------|---------|---------|---------|
| t | 0. | 0.5 | 1.0 | 1.5 | 2.0 | 2.5 |
| $h = 0.5$ | 1. | 1.61486 | 2.31919 | 2.71076 | 2.48190 | 1.81876 |
| $h = 0.25$ | 1. | 1.61513 | 2.31974 | 2.71144 | 2.48254 | 1.81931 |
| $g(t)$ | 1. | 1.61515 | 2.31978 | 2.71148 | 2.48258 | 1.81934 |

Part IV. Linear Differential Equations

Section 1. Linear Oscillators

2.* Consider now the following damped system. Obtain the solution.

$$y'' + y' + 4y = 0, y(0) = 2, y'(0) = -3$$

From the form of the solution decide if the system is underdamped, critically damped, or overdamped.

```
> DE := diff(y(t),t,t) + diff(y(t),t) + 4*y(t) = 0;
    inits := y(0)=2, D(y)(0)=-3;
```

$$DE := \left(\frac{d^2}{dt^2} y(t) \right) + \left(\frac{d}{dt} y(t) \right) + 4y(t) = 0$$

$$inits := y(0) = 2, D(y)(0) = -3$$

```
> soln := dsolve( {DE,inits} );
```

$$soln := y(t) = -\frac{4}{15} \sqrt{15} e^{\left(-\frac{1}{2}t\right)} \sin\left(\frac{1}{2} \sqrt{15} t\right) + 2 e^{\left(-\frac{1}{2}t\right)} \cos\left(\frac{1}{2} \sqrt{15} t\right)$$

The system is underdamped.

a. What is the pseudo-period of the oscillations?

The pseudo-period is

```
> 2*Pi/(sqrt(15)/2): evalf[4](%)*time_units;
    3.246 time_units
```

b. What is the time constant?


The time constant is 2 time units.

c. Based upon your answer to part b estimate the time interval required for the oscillations to disappear from view.

The oscillations will disappear in about $5 \times 2 = 10$ time units.

d. Plot the solution curve over the interval you named in part c.

```
> plot( rhs(soln), t=0..10);
```




e. Add to the curve in part d the curves defined by $A\exp(-t/2)$ and $-A\exp(t/2)$ where

$$A = \sqrt{4 + \frac{16}{15}}.$$

Make them blue. What is the significance of these curves? Where did the formula for A come from?

```
> A := sqrt(4 + 16/15):
plot( [rhs(soln), A*exp(-t/2), -A*exp(-t/2)], t=0..10,
color=[red,blue$2]);
```




The blue curves are the envelopes for the oscillations.

The value of A is derived from the standard formula used to convert the sum of a sine and a cosine into a sinusoid.


f. Use unapply to convert the solution into the function g. Use g to plot the phase plane trajectory.

```
> g := unapply(rhs(soln), t):
plot( [g(t), D(g)(t), t=0..10] );
```




g. Add to the trajectory the points corresponding to $t = 0, 0.25, 0.5, 0.75, \dots, 2.0$.

```
> Points := [g(t/4),D(g)(t/4)] $ t=0..8:
  plot( [ [g(t),D(g)(t),t=0..10], [ Points ] ], style=[line,point],
        color=[red,black] );
```




4.* Use DEplot to draw the direction field in phase space for the damped system. Then add the solution trajectory.

```
> DEsystem := diff(y(t),t) = v(t), diff(v(t),t) = -v(t) - 4*y(t):
> DEtools[DEplot]( {DEsystem}, [y(t),v(t)], t=0..10, y=-2..2, v=-4..2,
  dirgrid=[11,11], arrows=slim);
```



```
> DEtools[DEplot]( {DEsystem}, [y(t),v(t)], t=0..10, y=-2..2, v=-4..2,
  dirgrid=[11,11], arrows=slim, {[y(0)=2,v(0)=-3]},
  linecolor=blue, stepsize=0.1);
```



Section 2. State Space

2.* Damp the system slightly by changing the IVP to the following

$$y'' + 0.1y' + 4y = \cos(1.8 t) , y(0) = 2 , y'(0) = -3$$

Obtain the solution, call it soln and convert it into a function g using unapply. Suppress the output for soln and the definition of g and then enter

```
evalf[3]( 'g(t)' = g(t) )
```

to see a nice looking representation of the solution formula.

```
> DE := diff(y(t),t,t) + 0.1*diff(y(t),t) + 4*y(t) = cos(1.8*t);
  inits := y(0)=2, D(y)(0)=-3;
```


$$DE := \left(\frac{d^2}{dt^2} y(t) \right) + 0.1 \left(\frac{d}{dt} y(t) \right) + 4 y(t) = \cos(1.8 t)$$

$$inits := y(0) = 2, D(y)(0) = -3$$

```
> soln := dsolve( {DE,inits} ):
g := unapply(rhs(soln),t):
evalf[3]('g(t)'=g(t));
g(t) = -1.75 e(-0.0500 t) sin(2.00 t) + 0.754 e(-0.0500 t) cos(2.00 t) + 1.25 cos(1.80 t) + 0.295 sin(1.80 t)
```

a. Based upon the solution formula determine the time constant for the beats. That is, how long will it take (approximately) for the beats to disappear from the solution curve as it settles down to its steady-state mode?


The time constant is $1/0.05$ time units.

```
> 1/0.05*time_units;
20.00000000 time_units
```

It will take approximately 100 seconds for the beats to disappear from the solution curve.


b. Plot the solution curve to verify your answer to part a.

```
> plot( g(t), t=0..100 );
```



c. Use g to obtain the phase plane and state space trajectories for this system.

```
> plot( [g(t),D(g)(t),t=0..100], title="Phase Plane");
```




```
> plots[spacecurve]( [g(t),D(g)(t),t,t=0..100], numpoints=600,
```

```

orientation=[-60,70], axes=boxed,
labels=["Position","Velocity","Time"],
color=red, title="State Space" );

```

State Space




5.* Apply DEsystem it to the differential equation in Exercise 2 in this section and use DEplot to obtain the solution trajectory.

```

> DEsystem := diff(y(t),t) = v(t),
               diff(v(t),t) = -0.1*v(t) - 4*y(t) + cos(1.8*t):
DEtools[DEplot]( {DEsystem}, [y(t),v(t)], t=0..100, y=-3..3, v=-6..5,
                 linecolor=blue, stepsize=0.1, {[y(0)=3,v(0)=-2]} ,
                 title="Phase Space");


```



6.* Continuing 5. Apply DEplot3d to obtain the state space trajectory for the IVP in Exercise 2.

```
> DETools[DEplot3d]( {DEsystem}, [y(t),v(t)], t=0..100, y=-3..3,
v=-6..5, scene=[y,v,t], orientation=[-60,70],
axes=boxed, labels=["Position","Velocity","Time"],
title="State Space" ,
linecolor=blue, stepsize=0.1, {[y(0)=3,v(0)=-2]} );
```

State Space



Section 3. Two Dimensional Systems

4.* The model


$$x' = x(1 - y - x/a)$$

$$y' = y(1 - x - y/b)$$

is used to study competing species. See Ledder, Section 5.5, Exercise 11. Use DEplot to do the following.


a. Draw the direction field when $1/a = 1.9$ and $1/b = 1.5$. Use the window $x = 0..1$, $y = 0..1$.

```
> with(DEtools):
> DEsystem := diff(x(t),t) = x(t)*(1 - y(t) - 1.9*x(t)),
               diff(y(t),t) = y(t)*(1 - x(t) - 1.5*y(t));
               DEsystem :=  $\frac{d}{dt}x(t) = x(t)(1 - y(t) - 1.9x(t))$ ,  $\frac{d}{dt}y(t) = y(t)(1 - x(t) - 1.5y(t))$ 
> DEplot( {DEsystem}, [x(t),y(t)], t=0..10, x=0..1, y=0..1,
           arrows=slim, dirgrid=[13,13]);
DF := %:
```



b. Find the stationary point and add the nullclines to the direction field. Discuss the solutions based upon the picture you see.

```
> stpt := solve( {1-y-1.9*x,1-x-1.5*y},{x,y}):
StatPoint := subs(stpt,[x,y]);
Nullclines := plots[implicitplot]( [1-y-1.9*x,1-x-1.5*y], x=0..1,
                                   y=0..1, color=green, thickness=2):
StatPoint := [0.2702702703, 0.4864864865]
> plots[display]( DF, Nullclines, plot( [StatPoint], style=point,
                                       color=black));
```




Solutions will stream towards the stationary point, moving as indicated in the four regions determined by the nullclines.

c. Add solution curves corresponding to the following set of initial conditions (a circle of points around the stationary point)

$$\{ [x(0)=0.3+0.2*\cos(\text{Pi}/6*k),y(0)=0.5+0.2*\sin(\text{Pi}*k/6)] \text{ } \$ k=0..11 \}$$

```
> DF := DEplot( {DEsystem}, [x(t),y(t)], t=-10..10, x=0..1, y=0..1,
  arrows=slim, dirgrid=[13,13], linecolor=blue,
  { [x(0)=0.3+0.2*cos(Pi/6*k),y(0)=0.5+0.2*sin(Pi*k/6)]$k=0..11 } ):
plots[display]( DF, Nullclines, plot( [StatPoint], style=point,
  color=black));
```



Section 4. Matrix Methods

3.* Define the three column vectors b1, b2, b3 as follows

$$b1,b2,b3 := \langle 0,1,-1 \rangle, \langle 1,1,0 \rangle, \langle -1,0,1 \rangle;$$

```
> with(LinearAlgebra):
```

> **b1,b2,b3 := <0,1,-1>,<1,1,0>,<-1,0,1>;**

$$b1, b2, b3 := \begin{bmatrix} 0 \\ 1 \\ -1 \end{bmatrix}, \begin{bmatrix} 1 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix}$$

a. Define the matrix B having b1, b2, b3 as its columns with the entry **B := <b1|b2|b3>**.

> **B := <b1|b2|b3>;**

$$B := \begin{bmatrix} 0 & 1 & -1 \\ 1 & 1 & 0 \\ -1 & 0 & 1 \end{bmatrix}$$

b. Find the characteristic polynomial of B and factor it with **factor(%)**.

> **CharacteristicPolynomial(B,t);**
factor(%);

$$t^3 - 2t^2 - t + 2 \\ (t-1)(t-2)(t+1)$$

c. Find B's eigenvalues with **Eigenvalues(B)**.

> **Eigenvalues(B);**

$$\begin{bmatrix} -1 \\ 1 \\ 2 \end{bmatrix}$$

d. Find B's eigenvectors using **Eigenvectors(B)**. Name the input **lambda,V**.

> **lambda,V := Eigenvectors(B);**

$$\lambda, V := \begin{bmatrix} 1 \\ 2 \\ -1 \end{bmatrix}, \begin{bmatrix} 0 & -1 & 2 \\ 1 & -1 & -1 \\ 1 & 1 & 1 \end{bmatrix}$$

e. Calculate **MatrixInverse(V).B.V**.

> **MatrixInverse(V).B.V;**

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & -1 \end{bmatrix}$$

4.* Continuing 3. Obtain the solution to $v' = Bv$ satisfying $v(0) = \langle 1, 2, 3 \rangle$.

a. Do it first using dsolve applied to the system of linear differential equations defined by $v' = Bv$ with the appropriate initial conditions.

> **u := <x(t), y(t), z(t)>:**

DEsystem := 'diff(u[k], t) = (B.u)[k]' \$ k=1..3;

$$DEsystem := \frac{d}{dt} x(t) = y(t) - z(t), \frac{d}{dt} y(t) = x(t) + y(t), \frac{d}{dt} z(t) = -x(t) + z(t)$$

> **soln := dsolve({DEsystem, x(0)=1, y(0)=2, z(0)=3});**

$$soln := \left\{ z(t) = \frac{1}{2} e^{-t} + \frac{5}{2} e^t, x(t) = e^{-t}, y(t) = -\frac{1}{2} e^{-t} + \frac{5}{2} e^t \right\}$$

> **VectorForm = subs(soln, u);**

$$VectorForm = \begin{bmatrix} e^{-t} \\ -\frac{1}{2} e^{-t} + \frac{5}{2} e^t \\ \frac{1}{2} e^{-t} + \frac{5}{2} e^t \end{bmatrix}$$

> **Check;**

eval(%%, t=0);

eval(DEsystem, soln);

Check

$$VectorForm = \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

$$-e^{-t} = -e^{-t}, \frac{1}{2} e^{-t} + \frac{5}{2} e^t = \frac{1}{2} e^{-t} + \frac{5}{2} e^t, -\frac{1}{2} e^{-t} + \frac{5}{2} e^t = -\frac{1}{2} e^{-t} + \frac{5}{2} e^t$$

b. Do it second by making a fundamental matrix solution $X(t)$ defined as the matrix with the eigenvector solutions in the columns then computing

$$v(t) = X(t) X(0)^{-1} v(0)$$

> **for k from 1 to 3 do W[k] := exp(lambda[k]*t)*Column(V, k): end do:**

```

unassign('k');
> <W[1]|W[2]|W[3]>;
X := unapply(%,t):

```

$$\begin{bmatrix} 0 & -e^{(2t)} & 2e^{(-t)} \\ e^t & -e^{(2t)} & -e^{(-t)} \\ e^t & e^{(2t)} & e^{(-t)} \end{bmatrix}$$

```

> v := X(t).X(0)^(-1).<1,2,3>;

```

$$v := \begin{bmatrix} e^{(-t)} \\ -\frac{1}{2}e^{(-t)} + \frac{5}{2}e^t \\ \frac{1}{2}e^{(-t)} + \frac{5}{2}e^t \end{bmatrix}$$

```

> Check;
eval(v,t=0);
map(diff,v,t) = B.v;

```

Check

$$\begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

$$\begin{bmatrix} -e^{(-t)} \\ \frac{1}{2}e^{(-t)} + \frac{5}{2}e^t \\ -\frac{1}{2}e^{(-t)} + \frac{5}{2}e^t \end{bmatrix} = \begin{bmatrix} -e^{(-t)} \\ \frac{1}{2}e^{(-t)} + \frac{5}{2}e^t \\ -\frac{1}{2}e^{(-t)} + \frac{5}{2}e^t \end{bmatrix}$$

c. To it third by using the Matrix Exponential, Exp_At . Once you have it, the solution is

$$v(t) = \text{Exp_At} v(0)$$

```

> Exp_At := MatrixExponential(B,t);

```


$$\text{Exp_At} := \begin{bmatrix} \frac{2}{3} e^{(-t)} + \frac{1}{3} e^{(2t)} & \frac{1}{3} e^{(2t)} - \frac{1}{3} e^{(-t)} & -\frac{1}{3} e^{(2t)} + \frac{1}{3} e^{(-t)} \\ \frac{1}{3} e^{(2t)} - \frac{1}{3} e^{(-t)} & \frac{1}{6} e^{(-t)} + \frac{1}{3} e^{(2t)} + \frac{1}{2} e^t & -\frac{1}{3} e^{(2t)} + \frac{1}{2} e^t - \frac{1}{6} e^{(-t)} \\ -\frac{1}{3} e^{(2t)} + \frac{1}{3} e^{(-t)} & -\frac{1}{3} e^{(2t)} + \frac{1}{2} e^t - \frac{1}{6} e^{(-t)} & \frac{1}{6} e^{(-t)} + \frac{1}{3} e^{(2t)} + \frac{1}{2} e^t \end{bmatrix}$$

> **v := Exp_At.<1,2,3>;**

$$v := \begin{bmatrix} e^{(-t)} \\ -\frac{1}{2} e^{(-t)} + \frac{5}{2} e^t \\ \frac{1}{2} e^{(-t)} + \frac{5}{2} e^t \end{bmatrix}$$

> **Check;**
See_above;

Check
See_above

d. Check the solution in each case.

Section 5. The Laplace Transform

6.* Consider the following initial value problem

$$y'' + y' + 9y = H(t - 2\pi) - H(t - 5\pi) + 2\text{Dirac}(t - 9\pi), \quad y(0) = 0, \quad y'(0) = 1$$

a. Obtain the solution using dsolve. Plot it for $t = 0..40$ and explain the behavior of the solution curve.

> **alias(H=Heaviside,delta=Dirac);**

H, δ

> **DE := diff(y(t),t,t) + diff(y(t),t) + 9*y(t) =**
H(t-2*Pi)-H(t-5*Pi)+2*delta(t-9*Pi);
inits := y(0)=0, D(y)(0)=1;


$$DE := \left(\frac{d^2}{dt^2} y(t) \right) + \left(\frac{d}{dt} y(t) \right) + 9y(t) = H(t - 2\pi) - H(t - 5\pi) + 2\delta(t - 9\pi)$$

inits := y(0) = 0, D(y)(0) = 1

> **soln := dsolve({DE,inits});**


$$\begin{aligned}
\text{soln} := y(t) = & \frac{2}{35} e^{\left(-\frac{1}{2}t\right)} \sin\left(\frac{1}{2}\sqrt{35}t\right) \sqrt{35} - \frac{1}{9} H(t-2\pi) e^{\left(-\frac{1}{2}t+\pi\right)} \cos\left(\frac{1}{2}\sqrt{35}t - \sqrt{35}\pi\right) \\
& - \frac{1}{315} \sqrt{35} H(t-2\pi) e^{\left(-\frac{1}{2}t+\pi\right)} \sin\left(\frac{1}{2}\sqrt{35}t - \sqrt{35}\pi\right) \\
& + \frac{1}{9} H(t-5\pi) e^{\left(-\frac{1}{2}t+\frac{5}{2}\pi\right)} \cos\left(\frac{1}{2}\sqrt{35}t - \frac{5}{2}\sqrt{35}\pi\right) \\
& + \frac{1}{315} \sqrt{35} H(t-5\pi) e^{\left(-\frac{1}{2}t+\frac{5}{2}\pi\right)} \sin\left(\frac{1}{2}\sqrt{35}t - \frac{5}{2}\sqrt{35}\pi\right) \\
& + \frac{4}{35} \sqrt{35} H(t-9\pi) e^{\left(-\frac{1}{2}t+\frac{9}{2}\pi\right)} \sin\left(\frac{1}{2}\sqrt{35}t - \frac{9}{2}\sqrt{35}\pi\right) + \frac{1}{9} H(t-2\pi) - \frac{1}{9} H(t-5\pi)
\end{aligned}$$

```
> plot( rhs(soln), t=0..40);
```



The solution curve shows the response to the steady force from $t = 6$ to $t = 16$ (roughly). See the plot of the driver below. Then there is no force until the kick provided by the Dirac delta at about $t = 27$. Note that at $t = 27$ the system reacts as it did at $t = 0$ except it goes twice as far from equilibrium.

```
> plot( rhs(DE), t=0..40, title="Heaviside in Driver");
```



b. Obtain the solution using `dsolve/method=laplace`. How does the solution formula compare to the formula obtained in part a?

```
> soln2 := dsolve( {DE,inits}, y(t), method=laplace );
```

$$\begin{aligned}
\text{soln2} := y(t) = & \frac{2}{35} e^{\left(-\frac{1}{2}t\right)} \sin\left(\frac{1}{2}\sqrt{35}t\right) \sqrt{35} \\
& + \frac{4}{35} H(t-9\pi) \sqrt{35} e^{\left(-\frac{1}{2}t + \frac{9}{2}\pi\right)} \sin\left(\frac{1}{2}\sqrt{35}(t-9\pi)\right) + \frac{1}{630} \left(70 \right. \\
& \left. - I(-35I + \sqrt{35}) e^{\left(-\frac{1}{2}I(-I + \sqrt{35})(t-2\pi)\right)} + I e^{\left(\frac{1}{2}I(I + \sqrt{35})(t-2\pi)\right)} (35I + \sqrt{35}) \right) t \\
& - 2\pi + \frac{1}{630} \left(-70 + I(-35I + \sqrt{35}) e^{\left(-\frac{1}{2}I(-I + \sqrt{35})(t-5\pi)\right)} \right. \\
& \left. - I e^{\left(\frac{1}{2}I(I + \sqrt{35})(t-5\pi)\right)} (35I + \sqrt{35}) \right) H(t-5\pi)
\end{aligned}$$


The Laplace solution is in complex form, probably because of a factorization of the characteristic polynomial over the complex field.

7.* Use `unapply` to convert the solution to 6 a into a function, `g`. Use `g` to plot the phase plane trajectory and as well as the state space trajectory for the system. Use the following options in the `spacecurve` procedure for the state space trajectory

```
axes=boxed, orientation=[-60,70], color=red, numpoints=600,
labels=["Position","Velocity","Time"]
```


```
> g := unapply(rhs(soln), t):
```

```
> plot( [g(t),D(g)(t),t,t=0..100], title="Phase Space Trajectory");
```



```
> plots[spacecurve]( [g(t),D(g)(t),t,t=0..100],
title="State Space Trajectory",
axes=boxed, orientation=[-60,70], color=red,
numpoints=800,
labels=["Position","Velocity","Time"] );
```

State Space Trajectory



8.* Continuing 7. Use the entry

```
plots[animate]( plot, [ [g(t),D(g)(t),t=0..T] ], T=0..40, frames=82);
```

to plot an animation of the phase space trajectory. Click on the plot and use the video controls in the context bar to play the animation and also to view it one frame at a time. Compare the trajectory movement to the time series for position.

Note. The slide control can also be used to control the display of the frames. Assuming t is measured in seconds, you can reduce the frame rate to 2 frames per second to view the trajectory in "real time".

```
> plots[animate]( plot, [ [g(t),D(g)(t),t=0..T] ], T=0..40, frames=82);  
T = .49383
```

